LBID-462
UC 38

Lawrence Berkeley Laboratory
UNIVERSITY OF CALIFORNIA

Engineering & Technical Services Division



DISCLAIMER

LAWRENCE BERKELEY LABORATORY - UNIVERSITY OF CALIFORNIA ENGINEERING NOTE	CODE MD1111	SERIAL M 5826	PAGE 1 OF 14
AUTHOR R. MEUSER	DEPARTMENT MECH. ENG.	LOCATION BERK	DATE DEC 3, 1981

PROGRAM — PROJECT — JOB

HIGH-FIELD MAGNET DEVELOPMENT

ANALYSIS

TITLE

INTERSECTING ELLIPSE COIL GENERATOR

The program presented generates a set of horizontal rectangular current blocks that closely approximates in "intersecting ellipse" dipole-magnet cross section. It is in subroutine form and runs on the HP 9845B desktop computer. It seems to work pretty well. Here is how it works:

One inputs the semi-width and aspect ratio, width / height, of an "aperture ellipse", and the outside semi-width of the coil. The program generates a pair of intersecting ellipses which are coincident with and have the same radius of curvature as the aperture ellipse at its intersections with the x-axis, and which give the correct overall width. The user specifies the overall height of the stack of current blocks, and the number of blocks. The program determines the width and horizontal position of each block in a manner described later. The total cross section of the set of blocks equals the cross section between the intersecting ellipses. Here's an example.



Nifty, eh?

Keyboard inputs requested by main program
See next page for nomenclature

Prob$ Problem identification code
R
Ar
D
Ans1$ = "Y" if Dtot = B
= "N" if Dtot ≠ B
Dtot (requested only if Ans1$ = "N")
Nd
Scale Scale for plotting, mm per unit of input data. (If dimensions are inputted in mm and Scale is inputted 0.5, a half-scale drawing is made.)

Outputs: see typical output, end of this report.





NOMENCLATURE FOR COMPUTER PROGRAM



Ground rule:
For the simplest case where $y_2 < Y_m$,

Make x_1 and x_2 such that $A_1' = A_1$ and $A_2' = A_2$

Non-simplest cases are illustrated below.



The total area of all blocks, then, equals the total area between the ellipses.

There are 11 cases to be considered depending on whether y_1 or y_2 are $=$, $<$, or $>$ y_m or b, and in addition there is the special consideration where $y_2 < b$ for the uppermost block. The 11 cases are represented below



Brackets represent limits of block

Case	y_1	y_2
1	$< y_m$	$< y_m$
2	$< y_m$	$= y_m$
3	$< y_m$	$> y_m$ and $< b$
4	$< y_m$	$= b$
5	$< y_m$	$> b$
6	$= y_m$	$< b$
7	$= y_m$	$= b$
8	$= y_m$	$> b$
9	$> y_m$	$< b$
10	$> y_m$	$= b$
11	$> y_m$	$> b$

In general, $y_2 > y_1$

We can combine the cases for y_1 or $y_2 = y_m$ or b with others:

Case		
1, 2	$< y_m$	$\leq y_m$
3, 4, 6, 7	$\leq y_m$	$> y_m$ and $\leq b$
5, 8	$\leq y_m$	$> b$
9, 10	$> y_m$	$\leq b$
11	$> y_m$	$> b$

Cases 1, 2:

Calculate A using y_1, y_2 and

$$\left|\begin{matrix} x_1 \\ x_2 \end{matrix}\right| = \frac{a}{b}\sqrt{b^2 - \left|\begin{matrix} y_1 \\ y_2 \end{matrix}\right|^2},$$

then

$$\left|\begin{matrix} Xx_1 \\ Xx_2 \end{matrix}\right| = \frac{A}{y_2 - y_1} \mp C$$



Cases 3, 4, 6, 7:



Calculate A_1 after replacing y_2 by y_m
" A_2 " " y_1 by y_m

Then $A_3 = A_1 + A_2$

$$A_4 = 2c(y_2 - y_m) - A_2$$

$$Xx_2 = \frac{A_3}{y_2 - y_1} + C$$

$$Xx_1 = \frac{A_1 + A_4}{y_2 - y_1} - C = \frac{A_1 + 2c(y_2 - y_m) - A_2}{y_2 - y_1} - C$$

Cases 5, 8



Calculate A_1 after replacing y_2 by y_m
" A_2 " " y_1 by y_m and y_2 by b

$$A_3 = A_1 + A_2$$
$$A_4 = 2c(y_m - y_1) - A_1$$
$$f = \frac{A_3}{y_2 - y_1}, \quad g = \frac{A_2 + A_4}{y_2 - y_1}$$
$$Xx_1 = c - g, \quad Xx_2 = c + f, \text{ or}$$
$$Xx_1 = c - \frac{A_2 + 2c(y_m - y_1) - A_1}{y_2 - y_1}$$
$$Xx_2 = c + \frac{A_1 + A_2}{y_2 - y_1}$$

Cases 9, 10

Calculate A using x_1, x_2, y_1, y_2, then

$$\left|\begin{matrix} Xx_1 \\ Xx_2 \end{matrix}\right| = c \mp \frac{A}{y_2 - y_1}$$

Case 11

Calculate A after replacing y_2 by b, then

$$\left|\begin{matrix} Xx_1 \\ Xx_2 \end{matrix}\right| = c \mp \frac{A}{y_2 - y_1}$$

Special consideration for uppermost block

If y_2 of the uppermost block $< b$, then

Calculate A after replacing y_2 by b.
Calculate X_{x_1}, X_{x_2} using actual y_2

Applies to cases 3, 6, 9.

```
16250 ! PROGRAM ELIPS1, MEUSER, NOV. 1981.**********************
16251 PRINTER IS 0
16270 Start1: !
16280 INPUT "PROBLEM IDENTIFICATION CODE = ???",Prob$
16290 INPUT "APERTURE RADIUS = ???",R
16300 INPUT "WIDTH/HEIGHT RATIO OF APERTURE ELLIPSE = ???",Ar
16310 INPUT "RADIUS TO OUTSIDE OF COIL = ???",D
16320 A=.5*(D+R)
16330 C=.5*(D-R)
16340 Rcurv=R/Ar^2
16350 H=R/Ar
16360 B=SQR(Rcurv*A)
16370 Ym=B/A*SQR(A*A-C*C)
16380 Start2: !
16390 DISP "HEIGHT OF COIL IS  ",B
16400 INPUT "IS HEIGHT OF STACK TO BE SAME AS HEIGHT OF COIL (Y OR N) ???",Ans1$
16410 IF Ans1$="N" THEN 16450
16420 IF Ans1$<>"Y" THEN 16400
16430 Dtot=B
16440 GOTO 16470
16450 INPUT "OVERALL HEIGHT OF STACK = ???",Dtot
16460 DISP "MIN NUMBER IF LAYERS IS ",Dtot/(Dtot-Ym)
16470 Nd=NUM(1) INPUT "NUMBER OF LAYERS = ???",Nd
16480 PRINT USING "@"
16490 PRINT "PROGRAM USED: DERIVED FROM ELIPS1, DISC 8"
16500 PRINT "PROBLEM: ",Prob$
16510 PRINT "APERTURE RADIUS                          R = ",R
16520 PRINT "APERTURE ASPECT RATIO, WIDTH/HEIGHT Ar= ",Ar
16530 PRINT "APERTURE RAD. OF CURV. ON X-AXIS Rcurv= ",Rcurv
16540 PRINT "RADIUS TO OUTSIDE OF COIL                D = ",D
16550 PRINT "COIL THICKNESS                   E = D-R = ",D-R
16560 PRINT "SEMI MAJOR AXIS                          A = ",A
16570 PRINT "SEMI MINOR AXIS                          B = ",B
16580 PRINT "HORIZONTAL OFFSET OF VERT. AXIS          C = ",C
16590 PRINT "VERTICAL COORDINATE OF INTERSECTION"
16600 PRINT "   OF ELLIPSES WITH VERT. AXIS          Ym = ",Ym
16610 PRINT "OVERALL HEIGHT OF STACK               DTOT = ",Dtot
16620 PRINT "NUMBER OF LAYERS                        Nd = ",Nd
16630 Dlr=Dtot/Nd
16640 IF Dtot-Dlr<B THEN 16670
16650 DISP "INVALID INPUT. TRY AGAIN."
16660 GOTO Start2
16670 Y1=-Dlr
16680 FOR I=1 TO Nd
16690 Y1=Y1+Dlr
16700 Y2=Y1+Dlr
16710 Yy1(I)=Y1
16720 Yy2(I)=Y2
16730 IF (Y1<Ym) AND (Y2<Ym) THEN Cs(I)=1
16740 IF (Y1<Ym) AND (Y2=Ym) THEN Cs(I)=2
16750 IF (Y1<Ym) AND (Y2>Ym) AND (Y2<B) THEN Cs(I)=3
16760 IF (Y1<Ym) AND (Y2=B) THEN Cs(I)=4
16770 IF (Y1<Ym) AND (Y2>B) THEN Cs(I)=5
16780 IF (Y1=Ym) AND (Y2<B) THEN Cs(I)=6
16790 IF (Y1=Ym) AND (Y2=B) THEN Cs(I)=7
16800 IF (Y1=Ym) AND (Y2>B) THEN Cs(I)=8
16810 IF (Y1>Ym) AND (Y2<B) THEN Cs(I)=9
16820 IF (Y1>Ym) AND (Y2=B) THEN Cs(I)=10
```

```
16830 IF (Y1>Ym) AND (Y2>B) THEN Cs(I)=11
16840 ON Cs(I) GOTO C1,C1,C3,C3,C5,C3,C3,C5,C9,C9,C11
16850 C1: ! ************ Cs = 1 OR 2
16860 X1=A/B*SQR(B*B-Y1*Y1)
16870 X2=A/B*SQR(B*B-Y2*Y2)
16880 CALL Area(X1,X2,Y1,Y2,A,B,Area)
16890 Xx1(I)=Area/(Y2-Y1)-C
16900 Xx2(I)=Area/(Y2-Y1)+C
16910 GOTO Cont1
16920 C3: ! ************ Cs = 3, 4, 6, OR 7
16930 X1=A/B*SQR(B*B-Y1*Y1)
16940 CALL Area(X1,C,Y1,Ym,A,B,Area1)
16950 X2=A/B*SQR(B*B-Y2*Y2)
16951 Yy=Y2
16952 IF (I=Nd) AND (Y2<B) THEN Yy=B
16960 CALL Area(C,X2,Ym,Yy,A,B,Area2)
16970 Xx1(I)=(Area1-Area2+2*C*(Y2-Ym))/(Y2-Y1)-C
16980 Xx2(I)=(Area1+Area2)/(Y2-Y1)+C
16990 GOTO Cont1
17000 C5:! ************ Cs = 5 OR 8
17010 X1=A/B*SQR(B*B-Y1*Y1)
17020 CALL Area(X1,C,Y1,Ym,A,B,Area1)
17030 CALL Area(C,0,Ym,B,A,B,Area2)
17040 Xx1(I)=-(Area2-Area1+2*C*(Ym-Y1))/(Y2-Y1)+C
17050 Xx2(I)=(Area1+Area2)/(Y2-Y1)+C
17060 GOTO Cont1
17070 C9: ! ************ Cs = 9 OR 10
17080 X1=A/B*SQR(B*B-Y1*Y1)
17090 X2=A/B*SQR(B*B-Y2*Y2)
17091 Yy=Y2
17092 IF (I=Nd) AND (Y2<B) THEN Yy=B
17100 CALL Area(X1,X2,Y1,Yy,A,B,Area)
17110 Xx1(I)=-Area/(Y2-Y1)+C
17120 Xx2(I)=Area/(Y2-Y1)+C
17130 GOTO Cont1
17140 C11: ! ************ Cs = 11
17150 X1=A/B*SQR(B*B-Y1*Y1)
17160 CALL Area(X1,0,Y1,B,A,B,Area)
17170 Xx1(I)=-Area/(Y2-Y1)+C
17180 Xx2(I)=Area/(Y2-Y1)+C
17190 GOTO Cont1
17200 Cont1: ! **************
17210 NEXT I
17220 PRINT ""
17230 PRINT "  I       X1          X2       X2-X1         Y1          Y2          CASE"
17240 FOR I=1 TO Nd
17250 PRINT USING "DDD,5(2X,DDDD.DDD),5X,5D";I,Xx1(I),Xx2(I),Xx2(I)-Xx1(I),Yy1(I
),Yy2(I),Cs(I)
17260 NEXT I
17261 CALL R_min_max(Xx1(*),Xx2(*),Yy1(*),Yy2(*),Nd)
17270 X1=A/B*SQR(B*B-Ym*Ym)
17280 CALL Area(X1,0,Ym,B,A,B,Area)
17290 Area=2*C*Ym+2*Area
17300 PRINT "TOTAL AREA = ",Area
17310 Area=0
17320 FOR I=1 TO Nd
```

```
17330 Area=Area+(Xx2(I)-Xx1(I))*(Yy2(I)-Yy1(I))
17340 NEXT I
17350 PRINT "TOTAL AREA = ",Area
17360 Nblk(1)=Nd                                    !**********
17361 FOR I=1 TO Nd
17370 X11(1,I)=Xx1(I)                               !**********
17380 X22(1,I)=Xx2(I)                               !**********
17390 Y11(1,I)=Yy1(I)                               !**********
17400 Y22(1,I)=Yy2(I)                               !**********
17410 D(1)=Yy2(I)-Yy1(I)                           !**********
17420 W(1,I)=Xx2(I)-Xx1(I)                          !**********
17430 Nw(1,I)=INT(W(1,I)/Smax(1)-.00001)+1          !**********
17440 NEXT I                                        !**********
17450 MAT Ysp=ZER                                   !**********
17460 Nd(1)=INT(D(1)/Smax(1)-.00001)+1              !**********
17461 INPUT "DO YOU WANT A CROSS-SECTION DRAWING (Y OR N) ???",Ans85$
17462 IF Ans85$="Y" THEN 17470
17463 IF Ans85$="N" THEN SUBEXIT
17464 DISP "ILLEGAL INPUT. MUST BE 'Y' OR 'N'. TRY AGAIN."
17465 GOTO 17461
17470       ! DRAWS CROSS SECTION.
17480 PLOTTER IS "GRAPHICS"
17490 INPUT "SCALE = ???",Scl
17500 CALL Csc_setup(Prob$,Scl)
17510 CALL Csc_ellipse(A,B,C,R,H,Scl)
17520 CALL Csc_blocks(Nd,Xx1(*),Xx2(*),Yy1(*),Yy2(*),Scl,1)
17530 INPUT "DO YOU WANT A HARD COPY (Y OR N) ???",Anshc$
17540 IF Anshc$="N" THEN SUBEXIT
17550 IF Anshc$<>"Y" THEN 17530
17560 PRINT PAGE
17570 DUMP GRAPHICS
17580 GCLEAR
17590 EXIT GRAPHICS
17600 SUBEND !*****************************************************************
17610 SUB Area(X1,X2,Y1,Y2,A,B,Area)
17620 Area=.5*(X2*Y2-X1*Y1+A*B*(ASN(Y2/B)-ASN(Y1/B)))
17630 SUBEND !*****************************************************************
17640 SUB Csc_setup(Prob$,Scl)
17650 GRAPHICS
17660 Scl$=VAL$(Scl)
17670 LINE TYPE 1
17680 ! CENTERLINES
17690 FRAME
17700 MSCALE 0,0
17710 ! CENTERLINES
17720 X=-50
17730 FOR I=1 TO 8
17740 X=X+30
17750 MOVE X+2,10
17760 DRAW X+28,10
17770 MOVE X+29,10
17780 DRAW X+31,10
17790 NEXT I
17800 Y=-50
17810 FOR I=1 TO 6
17820 Y=Y+30
17830 MOVE 10,Y+2
17840 DRAW 10,Y+28
17850 MOVE 10,Y+29
17860 DRAW 10,Y+31
```

```
17870 NEXT I
17880 ! LABELS
17890 LORG 1
17900 MOVE 12,2
17910 LABEL "PROBLEM: "&Prob$
17920 MOVE 140,2
17930 LABEL "SCALE: "&Scl$
17940 SUBEND!*************************************************************
17950 SUB Csc_blocks(Nbltot,X1(*),X2(*),Y1(*),Y2(*),Scl,Ln)
17960 GRAPHICS
17970 LINE TYPE Ln
17980 Sc=Scl
17990 FOR I=1 TO Nbltot
18000 Xx1=X1(I)*Sc+10
18010 Xx2=X2(I)*Sc+10
18020 Yy1=Y1(I)*Sc+10
18030 Yy2=Y2(I)*Sc+10
18040 MOVE Xx1,Yy1
18050 DRAW Xx2,Yy1
18060 DRAW Xx2,Yy2
18070 DRAW Xx1,Yy2
18080 DRAW Xx1,Yy1
18090 NEXT I
18100 SUBEND !*****************************************************
18110 SUB Csc_ellipse(A,B,C,Rr,Hh,Scl)
18120 LINE TYPE 1
18130 Sc=Scl
18140 Aa=A*Sc
18150 Bb=B*Sc
18160 Cc=C*Sc
18170 R=Rr*Sc
18180 H=Hh*Sc
18190 X1old=Aa-Cc
18200 X2old=Aa+Cc
18210 X3old=R
18220 Yold=0
18230 Y3old=0
18240 DEG
18250 FOR Ang=2 TO 180 STEP 2
18260 Sn=SIN(Ang)
18270 Cs=COS(Ang)
18280 X1=Aa*Cs-Cc
18290 X2=X1+Cc+Cc
18300 X3=R*Cs
18310 Y=Bb*Sn
18320 Y3=H*Sn
18330 MOVE X1old+10,Yold+10
18340 DRAW X1+10,Y+10
18350 MOVE X2old+10,Yold+10
18360 DRAW X2+10,Y+10
18370 MOVE X3old+10,Y3old+10
18380 DRAW X3+10,Y3+10
18390 IF (X1<-10) AND (X2<-10) AND (X3<-10) THEN 18460
```

```
18400 X1old=X1
18410 X2old=X2
18420 X3old=X3
18430 Yold=Y
18440 Y3old=Y3
18450 NEXT Ang
18460 RAD
18470 SUBEND !*************************************************
19000 SUB R_min_max(X1(*),X2(*),Y1(*),Y2(*),Nblk)
19010 Rmin=1E12
19020 Rmax=0
19030 FOR I=1 TO Nblk
19040 Rin=SQR(X1(I)^2+Y1(I)^2)
19050 Rout=SQR(X2(I)^2+Y2(I)^2)
19060 IF Rin>Rmin THEN 19090
19070 Rmin=Rin
19080 Imin=I
19090 IF Rout<Rmax THEN 19120
19100 Rmax=Rout
19110 Imax=I
19120 NEXT I
19140 PRINT ""
19141 PRINT USING 19150;Rmin,Imin
19142 PRINT USING 19160;Rmax,Imax
19150 IMAGE "MIN. RADIUS = ",DDD.DDDDD,"  FOR BLOCK ",DD
19160 IMAGE "MAX. RADIUS = ",DDD.DDDDD,"  FOR BLOCK ",DD
19170 SUBEND !*************************************************************
```

TYPICAL PRINTOUT

```
PROGRAM USED: DERIVED FROM ELIPS1, DISC 8
PROBLEM:                1
APERTURE RADIUS                              R =   100
APERTURE ASPECT RATIO, WIDTH/HEIGHT         Ar=   1
APERTURE RAD. OF CURV. ON X-AXIS          Rcurv=  100
RADIUS TO OUTSIDE OF COIL                    D =   200
COIL THICKNESS                         E = D-R =   100
SEMI MAJOR AXIS                              A =   150
SEMI MINOR AXIS                              B =   122.474487139
HORIZONTAL OFFSET OF VERT. AXIS              C =   50
VERTICAL COORDINATE OF INTERSECTION
   OF ELLIPSES WITH VERT. AXIS              Ym =   115.470053838
OVERALL HEIGHT OF STACK                   DTOT =   122.474487139
NUMBER OF LAYERS                             Nd =   4
```

Dimensions as generated from intersecting ellipses (dimensions in mm)

I	X1	X2	X2-X1	Y1	Y2	CASE
1	98.423	198.423	100.000	0.000	30.619	1
2	88.561	188.561	100.000	30.619	61.237	1
3	66.259	166.259	100.000	61.237	91.856	1
4	25.533	117.997	92.464	91.856	122.474	4

```
MIN. RADIUS =  90.22314  FOR BLOCK  3
MAX. RADIUS = 200.77100  FOR BLOCK  1
TOTAL AREA =          12016.7053714
TOTAL AREA =          12016.7053713
```



TECHNICAL INFORMATION DEPARTMENT
LAWRENCE BERKELEY LABORATORY
UNIVERSITY OF CALIFORNIA
BERKELEY, CALIFORNIA 94720